
5.15 Kernelの基礎(2) 　　　　　　　　　　　　　　   ver.0.1

引き続き Kernel の基礎についてプログラムします。ここではまず Bivariate の場合の

Kernel Densityについてです。まず Bivariate Normalをシミュレーションするためにデー

タの組数ｎを指定してから、Variance Covariance行列を設定します。この場合の平均は２

変数ともに便宜上０にしてあります。ここで、x=rndn(n,2)*chol(vc);とあるのは、２列の標

準正規乱数を生成したものにVariance Covariance行列をCholesky分解したものをかけて

います。Cholesky分解は行列のルートのようなものですから、これをかけているというの

は Variance Covariance 行列のルートしたようなものをかけていると考えてください。ち

ょうど、σ＝２であれば rndn(n,1)*sigma とすることによって N(0,4)にしたがう乱数が作

成できるのと要領は同じです。そこに Covarianceが加わっているので Cholesky分解で行

列を変換しているだけです。これによってシミュレートされたデータを plotbikern でグラ

フ化することにします。なお、この場合のデータは Standardizedされていないものを使用

してそのまま計算するものとします。

プログラム

new; cls;

library pgraph;

graphset;

n=50;

vc={2    0.7,

    0.7  2  };

rndseed 1;

x=rndn(n,2)*chol(vc);

call plotbikern(x,vc);

proc plotbikern(x,vc);

   local point1,point2,points,i,j,dens,kern,x0;

   point1=seqa(minc(minc(x[.,1]),(maxc(x[.,1])-minc(x[.,1]))/98),99);

   point2=seqa(minc(minc(x[.,2]),(maxc(x[.,2])-minc(x[.,2]))/98),99);

   points=point1~point2;

   i=1; dens=zeros(99,99); kern=zeros(99,99);

   do while i<=99;

      j=1;

      do while j<=99;

         x0=points[i,1]~points[j,2];

         dens[i,j]=inv(2*pi)*inv(sqrt(det(vc)))*exp(-0.5*x0*inv(vc)*x0');



         kern[i,j]=kerneldens(x0,x);

         j=j+1;

      endo;

      i=i+1;

   endo;

/* Plot graphs */

   begwind;

   window(2,2,0);

   setwind(1);

      graphset;

      title("True Bivariate Normal Density");

      surface(point1',point2,dens);

   setwind(2);

      graphset;

      contour(point1',point2,dens);

   setwind(3);

      graphset;

      title("Kernel Density");

      surface(point1',point2,kern);

   setwind(4);

      graphset;

      contour(point1',point2,kern);

   endwind;

   retp(kern);

endp;

proc kerneldens(x0,x);

   local n,k,h0,hh,invs,dets,u,kn;

   n=rows(x);

   h0=1; /* ho depends on the type of kernels. */

   hh=h0*n^(-1/6);

   invs=inv(chol(vcx(x)));

   dets=sqrt(det(vcx(x)));

   u=(x0-x)./hh;

   kn=kernels(u*invs,"gauss");

   retp(sumc(kn)/(n*((hh^2)*dets)));



endp;

proc kernels(u,types);

   local kv;

   if types$=="uniform";

      kv=0.5*(abs(u).<=1);

   elseif types$=="gauss";

      kv=(1/sqrt(2*pi))*exp(-0.5*u^2);

   elseif types$=="triangular";

      kv=(1-abs(u)).*(abs(u).<=1);

   elseif types$=="biweight";

      kv=(15/16)*((1-u^2)^2).*(abs(u).<=1);

   elseif types$=="triweight";

      kv=(35/32)*((1-u^2)^3).*(abs(u).<=1);

   elseif types$=="epanechnikov";

      kv=0.75*(1-u^2).*(abs(u).<=1);

   else;

      errorlog "ERROR:  Type must be either of the following:";

      errorlog "uniform/gauss/triangular/biweight/triweight/epanechnikov";

   endif;

   retp( prodc(kv') );

endp;

上のプログラムでは、Univariateのケースとは違って、Kernelを計算する一番下の計算の

リターンが kvそのものではなくて prodc(kv’)としています。これは２次元であるため、こ

うなっています。さらに、その上の procedureの計算では、sumc(kn)/(n*((hh^2)*dets))を

リターンになっており、ｈｈ＾２と２乗になります。評価点は x0 で Univariate と変更は

ありません。Kernelの計算を procedureで呼び出すところで kn=kernels(u*invs,"gauss");

というふうに invs=inv(chol(vcx(x)));でかけているのは、standardizedをここでしているこ

とになります。１変数であればσで割ることに相当します。また、ここではバンド幅を h0

をまず１に止めておいて、hh=h0*n^(-1/6);によって最終的なバンド幅を概算で決定してい

ます。ここで(-1/6)というのは、-0.2 が１変数のときで、この場合２変数ですから５に１が

加わって、都合６の(-1/6)となっています。これをもとに、一番上のところで、True Density

は dens[i,j]=inv(2*pi)*inv(sqrt(det(vc)))*exp(-0.5*x0*inv(vc)*x0');により計算して、Kernel

の計算 kern[i,j]=kerneldens(x0,x);とともに 99×99 のグリッド上の高さを計算しているこ

とになります。なお、この 99×99のグリッドの大きさは任意なのですが、それぞれの軸の

最小値と最大値を求め、その間を 98分割して 99×99のグリッドを作成しています。なお、



contour グラフは奇数グリッドを必要とするので（偶数ではエラーになります）99 にとど

めています。また、ｘ軸の方を列から行に転置することで、ｙ軸とは変えることによって、

グリッド座標が自動的に与えられることはグラフィックスのところで行なったのと同じ要

領です。そうすると、下のように上段には True Densityの３次元グラフとその等高線。下

段には、Kernel Densityとその等高線が描かれます。

グラフ表示

Kernel Regression(２次元のケース)

ここでは、また Univariate のケースに立ち戻って、それを用いた Regression を考えま

す。定数項と１独立変数からなるモデルを考えて、まずは通常の OLSで計算できる回帰直

線 yhatと Kernel Regressionでもとめられた ghatとをプロットするプログラムを考えま

す。線形モデルｙ＝β0＋β1Ｘ＋uに対して、乱数設定を次のようにします。

　　　ｎ　　　β0　   β1　   Ｘ  　   u

　　1000      1       4     [0,5]    N(0,1)

プログラム

new; cls;

library pgraph;

graphset;

rndseed 1000;



/* Simulation */

   n=100;

   x=5*rndu(n,1); u=rndn(n,1); b={1,4};

   y=(ones(n,1)~x)*b+u;

/* Regression */

   ghat=kernelreg(y,x);

   x1=ones(n,1)~x;

   yhat=x1*(inv(x1'x1)*x1'y);

/* Plot */

   data=sortc(x~ghat~yhat~y,1);

   _plctrl={0,0,-1};

   _psymsiz=1;

   xy(data[.,1],data[.,2:4]);

proc kernelreg(y,x);

   local n,k,h0,h,invs,i,u,kv,w,ghat;

   n=rows(x); k=cols(x);

   h0=1; /* h0 depends on the type of kernels. */

   h=h0*n^(-1/(k+4));

   invs=inv(chol(vcx(x)));

   i=1; ghat=zeros(n,1);

   do while i<=n;

      u=(x[i]-x)./h;

      kv=kernels(u*invs,"epanechnikov");

      w=kv./sumc(kv);

      ghat[i]=w'y;

      i=i+1;

   endo;

   retp(ghat);

endp;

proc kernels(u,types);

   local kv;

   if types$=="uniform";

      kv=0.5*(abs(u).<=1);

   elseif types$=="gauss";



      kv=(1/sqrt(2*pi))*exp(-0.5*u^2);

   elseif types$=="triangular";

      kv=(1-abs(u)).*(abs(u).<=1);

   elseif types$=="biweight";

      kv=(15/16)*((1-u^2)^2).*(abs(u).<=1);

   elseif types$=="triweight";

      kv=(35/32)*((1-u^2)^3).*(abs(u).<=1);

   elseif types$=="epanechnikov";

      kv=0.75*(1-u^2).*(abs(u).<=1);

   else;

      errorlog "ERROR:  Type must be either of the following:";

      errorlog "uniform/gauss/triangular/biweight/triweight/epanechnikov";

   endif;

   retp( kv );

endp;

グラフ表示

上のグラフでは、赤い点がシミュレーションによって作成した実際データ。緑の実線が

Kernelで推定したもの。ブルーの破線がOLSによるものです。直線で推定されるOLSとは

違って、残差項の値の影響を受けてその都度曲がっていることがわかります。プログラム

において、kv=kernels(u*invs,"epanechnikov");でまず指定されたKernelタイプに対して、



uにinvsをかけたものインプットに使っていますが、この場合１変数（Kernelの場合定数項

は関係ない）なのでσで割っても同じことです。ここでは一般化しているだけです。次に、

w=kv./sumc(kv);のところでウエイトを計算しています。これを w’yとして１からｎまでル

ープを回すことで Kernel Regressionになります。

同じことは、非線形のデータについてもあてはめることができます（むしろ、こちらの

方が Kernel Regressionの適用分野と言えます）。ここでは、Polynomial y=2X3-X2+3X +u

に対して、乱数設定を次のようにするものとします。

　　　ｎ　　　 Ｘ  　    u

　 100     [-2,2]    N(0,22)

プログラム

new; cls;

library pgraph;

graphset;

rndseed 1000;

/* Simulation */

   n=100;

   x=4*rndu(n,1)-2; u=2*rndn(n,1);

   y=2*x^3-x^2+3*x+u;

   yhat=2*x^3-x^2+3*x;

/* Regression */

   ghat=kernelreg(y,x);

/* Plot */

   data=sortc(x~ghat~yhat~y,1);

   _plctrl={0,0,-1};

   _psymsiz=1;

   xy(data[.,1],data[.,2:4]);

proc kernelreg(y,x);

   local n,k,h0,h,invs,i,u,kv,w,ghat;

   n=rows(x); k=cols(x);

   h0=1; /* h0 depends on the type of kernels. */

   h=h0*n^(-1/(k+4));

   invs=inv(chol(vcx(x)));

   i=1; ghat=zeros(n,1);

   do while i<=n;



      u=(x[i]-x)./h;

      kv=kernels(u*invs,"epanechnikov");

      w=kv./sumc(kv);

      ghat[i]=w'y;

      i=i+1;

   endo;

   retp(ghat);

endp;

proc kernels(u,types);

   local kv;

   if types$=="uniform";

      kv=0.5*(abs(u).<=1);

   elseif types$=="gauss";

      kv=(1/sqrt(2*pi))*exp(-0.5*u^2);

   elseif types$=="triangular";

      kv=(1-abs(u)).*(abs(u).<=1);

   elseif types$=="biweight";

      kv=(15/16)*((1-u^2)^2).*(abs(u).<=1);

   elseif types$=="triweight";

      kv=(35/32)*((1-u^2)^3).*(abs(u).<=1);

   elseif types$=="epanechnikov";

      kv=0.75*(1-u^2).*(abs(u).<=1);

   else;

      errorlog "ERROR:  Type must be either of the following:";

      errorlog "uniform/gauss/triangular/biweight/triweight/epanechnikov";

   endif;

   retp( kv );

endp;

プログラムは冒頭の設定部分が違うだけで、procedeure 部分は線形のものと同じです。

Kernel Regressionは真のモデルがいかなるものであっても、あるいはモデルがそもそもな

いようのデータに対しても Regression が可能です。なお、上の設定では、Polynomial は

直接の推定が困難であるため、便宜上もとのモデルの乱数項 uがないものを yhatとしてい

ます。そうして、プロットすると次のようになります。



グラフ表示

Kernel Regression(３次元のケース)

　独立変数が定数項を勘定に入れないで２つの場合は、Bivariateと同じ手順で推定を行

ないます。以下では線形モデルｙ＝β0＋β1Ｘ1＋β2Ｘ2＋uに対して、乱数設定を

　　　ｎ　　　β0　    β1　   β2     Ｘ1     Ｘ2  　  u

　　 100      2        3      4      [0,5]    [0,5]    N(0,1)

として計算をします。

プログラム

new; cls;

library pgraph;

graphset;

rndseed 1000;

/* Simulation */

   n=100;

   x=5*rndu(n,2); u=rndn(n,1); b={2,3,4};

   y=(ones(n,1)~x)*b+u;

/* Regression & Graph */

   ghat=kernelreg(y,x);



proc kernelreg(y,x);

   local n,k,h0,h,invs,i,u,kv,w,ghat,point1,point2,points,x0,j,x1,bhat,yhat;

   n=rows(x); k=cols(x);

   h0=1; /* h0 depends on the type of kernels. */

   h=h0*n^(-1/(k+4));

   invs=inv(chol(vcx(x)));

   point1=seqa(minc(x[.,1]),(maxc(x[.,1])-minc(x[.,1]))/98,99);

   point2=seqa(minc(x[.,2]),(maxc(x[.,2])-minc(x[.,2]))/98,99);

   points=point1~point2;

   i=1; ghat=zeros(99,99);

   do while i<=99;

      j=1;

      do while j<=99;

         x0=points[i,1]~points[j,2];

         u=(x0-x)./h;

         kv=kernels(u*invs,"gauss");

         w=kv./sumc(kv);

         ghat[i,j]=w'y;

         j=j+1;

      endo;

      i=i+1;

   endo;

   x1=ones(n,1)~x;

   bhat=inv(x1'x1)*x1'y;

   i=1; yhat=zeros(99,99);

   do while i<=99;

      j=1;

      do while j<=99;

         x0=1~points[i,1]~points[j,2];

         yhat[i,j]=x0*bhat;

         j=j+1;

      endo;

      i=i+1;

   endo;

/* Plot panel graphs */

   begwind;



   window(2,2,0);

   setwind(1);

      surface(point1',point2,yhat);

   setwind(2);

      contour(point1',point2,yhat);

   setwind(3);

      surface(point1',point2,ghat);

   setwind(4);

      contour(point1',point2,ghat);

   endwind;

   retp(ghat);

endp;

proc kernels(u,types);

   local kv;

   if types$=="uniform";

      kv=0.5*(abs(u).<=1);

   elseif types$=="gauss";

      kv=(1/sqrt(2*pi))*exp(-0.5*u^2);

   elseif types$=="triangular";

      kv=(1-abs(u)).*(abs(u).<=1);

   elseif types$=="biweight";

      kv=(15/16)*((1-u^2)^2).*(abs(u).<=1);

   elseif types$=="triweight";

      kv=(35/32)*((1-u^2)^3).*(abs(u).<=1);

   elseif types$=="epanechnikov";

      kv=0.75*(1-u^2).*(abs(u).<=1);

   else;

      errorlog "ERROR:  Type must be either of the following:";

      errorlog "uniform/gauss/triangular/biweight/triweight/epanechnikov";

   endif;

   retp( prodc(kv') );

endp;

基本的に冒頭で行なった Bivariate の Kernel 計算と同じです。ここでも w=kv./sumc(kv);

としてウエイトを求めて、w’yを計算し Girdの数だけループを回しています。ただし、こ

この kvは一番下の Bivariateの procedureのリターン prodc(kv')のことです。



グラフ表示

グラフの上段は OLS による回帰面、下段は Kernel 回帰による面です。当然のことながら

線形の OLSによる回帰面の等高線は直線になります。

Least Square Jackknifeによる最適バンド幅の計算

　これまでは、バンド幅の概算を求めていました。今度はある統計基準を満たすバンド幅

を求めて、よりスムースなデータを反映した Kernel Rgressionになるようにしてみましょ

う。いろいろなやり方がありますが、一番単純な方法として、もとのデータｙとその Kernel

推定値 ghat との差の二乗和を計算するのに、Jackknife を用いて１つずつ取り除いたもの

を用いて、その Jackknife Least Squareを最小にするバンド幅ｈを選ぶ方法があります。

プログラム

new; cls;

library pgraph;

graphset;

rndseed 1000;

/* Simulation */

   n=100;

   x=10*rndu(n,1)-5; u=10*rndn(n,1);

   y=x^3+u;

/* Regression */



   h=linegrid(&f,0,10);

   ghat=kernelreg(y,x,h);

   yhat=x^3;

/* Plot */

   data=sortc(x~ghat~yhat~y,1);

   _plctrl={0,0,-1};

   _psymsiz=1;

   xy(data[.,1],data[.,2:4]);

   

proc f(h);

   local n,k,h0,h,invs,i,u,kv,w,ghat,x1,y1,index;

   n=rows(x); k=cols(x);

   i=1; ghat=zeros(n,1);

   do while i<=n;

      index=zeros(n,1);

      index[i]=1;

      u=(x[i]-x)./h;

      u=delif(u,index);

      x1=delif(x,index);

      invs=inv(chol(vcx(x1)));

      kv=kernels(u*invs,"triweight");

      w=kv./sumc(kv);

      y1=delif(y,index);

      ghat[i]=w'y1;

      i=i+1;

   endo;

   retp( -(y-ghat)'(y-ghat) );

endp;

proc kernelreg(y,x,h);

   local n,k,h0,h,invs,i,u,kv,w,ghat;

   n=rows(x); k=cols(x);

   invs=inv(chol(vcx(x)));

   i=1; ghat=zeros(n,1);

   do while i<=n;

      u=(x[i]-x)./h;



      kv=kernels(u*invs,"triweight");

      w=kv./sumc(kv);

      ghat[i]=w'y;

      i=i+1;

   endo;

   retp(ghat);

endp;

proc kernels(u,types);

   local kv;

   if types$=="uniform";

      kv=0.5*(abs(u).<=1);

   elseif types$=="gauss";

      kv=(1/sqrt(2*pi))*exp(-0.5*u^2);

   elseif types$=="triangular";

      kv=(1-abs(u)).*(abs(u).<=1);

   elseif types$=="biweight";

      kv=(15/16)*((1-u^2)^2).*(abs(u).<=1);

   elseif types$=="triweight";

      kv=(35/32)*((1-u^2)^3).*(abs(u).<=1);

   elseif types$=="epanechnikov";

      kv=0.75*(1-u^2).*(abs(u).<=1);

   else;

      errorlog "ERROR:  Type must be either of the following:";

      errorlog "uniform/gauss/triangular/biweight/triweight/epanechnikov";

   endif;

   retp( kv );

endp;

proc linegrid(&ll,start,finish);

   local length,maxiter,tol,i,j,step,grid,b,temp,bmax,ll:proc;

      length=100;

      maxiter=100;

      tol=1e-8;

   bmax=0;

   i=1;



   do while i<=maxiter;

       step=(finish-start)/length;

       grid=(-1e256).*ones(length+1,1);

       b=start;

       j=1;

       do while j<=length+1;

           grid[j]=ll(b);

           j=j+1;

           b=b+step;

        endo;

        temp=start+(maxindc(grid)-1)*step;

        if abs(bmax-temp)<tol;

           break;

        endif;

        bmax=temp;

        print/rd bmax;

        start=bmax-step; finish=bmax+step;

        i=i+1;

   endo;

   print/rz "# of iterations=" i-1;

   print/rd bmax;

   retp(bmax);

endp;

画面表示

      0.20000000

      0.25400000

      0.25312000

      0.25312400

      0.25312362

# of iterations=               5

      0.25312362



グラフ表示

上のように、自動的にｈは Grid(Line)Searchによって Jackknife Least Square f(h)を最小

にするｈをまず求めて、それを proc kernelreg(y,x,h)とｈを加えて３インプットに変更して

内部のｈの計算部分をなくした procedure にこの計算されたｈを受け渡しています。その

他の部分の Kernel Regressionはその前のものと同じです。ただし、f(h)を計算するのにｘ

とｙはインプットとしてではなくてグローバルに通しています。その１値関数 f(h)を以前に

作成した Grid(Line)Search のプログラムで関数ｆを＆をつけて呼び出して、０から 10 ま

での範囲を初期サーチの範囲として最小点を計算しています。中心となる f(h)は、kernelreg

をもう一回やったような形になっていますが、実際には、Jackknife された ghat を求める

ために、Jackknifeのところで行なったと全く同じように、まず０でできたｎ×１の index

を毎回作成して、そのｉ番目を１とします。それを論理と考えて、delif(列ベクトル,論理)

でｉ番目だけを切り落としています。この場合計算された uと invsを計算するためのｘ、

それにｙそのものの３つとも Jackknife で毎回切り落とします。そしてこれまでの kernel

regressionと同様に１からｎまでループを回して ghatを計算します。これともともとのｙ

との差の二乗和を計算して、この場合の Grid(Line)Searchの procedureは最大値を求める

ものだから、便宜上マイナスをつけたものをリターンにしておきます。なお、Jackknifeを

しないで Least Squareを計算したものを最小にしようとすると、ｈは限りなく０に近くな

り結果は全ての点を結ぶ軌跡になります。


